
11sstt  MMeeeettiinngg  ooff  EEccoossiimmPPrroo  UUsseerrss,,  UUNNEEDD,,  MMaaddrriidd,,  33--44  MMaayy  22000011

88--11

C8

DYNAMIC MODELLING AND SIMULATION WITH ECOSIMPRO
OF AN EVAPORATOR STATION IN THE SUGAR INDUSTRY

Esther Simón Pérez, Universidad de Valladolid
Real de Burgos s/n, 47011 Valladolid – Spain

essipe@terra.es

Abstract

This paper describes how a library of
evaporation process units created in EcosimPro
can be used to simulate the behaviour of any
evaporation station and experiment with
different working conditions.

For the tests, we have focused on an evaporator
station used in the sugar industry comprised of
five effects and eight evaporators. The first five
are of the Robert type and the last two of the
descending film type which allows us to build a
model which is very similar to the real
industrial facilities.

1 INTRODUCTION

Before concentrating on the development of the
evaporation library in EcosimPro, it is necessary
to explain some of the fundamental concepts of
evaporation to be able to correctly understand
the workings of a station. The objective of the
operation is to concentrate a dissolved solid into
a liquid.

The equipment in which the operation is carried
out is called an evaporator. It consists of a shell
where steam is condensed and a separation
chamber in which the solution to be
concentrated is boiled. As the solution looses
solvent, the proportion of solute increases.

Evaporation can be carried out in one or several
evaporators. When working with high flows and
the cost of steam is high, a chain of evaporators
is usually used. In this configuration the product
to be concentrated passes from one evaporator
to another in series, and the steam produced in
the evaporation of one of them is used for
heating the next evaporator. This is know as a

multiple effect system. The interest in this
multiple effect system basically lies in the more
efficient use of the steam. In these systems the
term first effect is given to the evaporators
which receive steam directly from the boilers,
the term second effect to those which receive
steam from the first effect, and so on. Likewise,
the steam extracted from each effect is called
vapour, so the vapour produced from the first
effect is called first effect vapour.

With regard to the flow of each of the fluids,
steam and solution to be concentrated, we have
the following:

- The steam flows from the effect with most
pressure to the effect of least pressure

- The solution to be concentrated can:
- Flow in the same direction as the

steam: direct flow
- Flow in the direction opposite to

that of the steam: countercurrent
flow

- Simultaneously feed several
evaporators: parallel flow

- Flow in no determined order:
mixed flow

The direct flow feed system is the most
common, largely because it is simpler to
operate. With this system, it is the difference in
pressure in each concentration chamber which
makes the solution to be concentrated flow
through the different evaporators. In some
cases, therefore, it is necessary to make a
vacuum in the last effect. The heat flux through
the condensation chamber (shell) and the
concentration chamber arises due to the
temperature difference between the two.

With regard to the selection of an evaporator
type, the trend is generally towards the one that
has the highest heat transfer coefficient in the
desired operating conditions. In this work we
have used two –the Robert types and the
descending film type- and so we will focus on
describing the two.

The Robert type evaporator has a vertical
cylindrical body and the tubes are anchored
between two horizontal tubular plates. It has a
large central downpipe with which a good
solution flow can be obtained and this is a big
advantage; however, its greatest disadvantage is
the amount of space it occupies.

Figure 1: Robert type evaporator

In the descending film evaporators the liquid to
be concentrated is pumped from a tank at the
bottom and flows into the top of the evaporator,
as does the vapour which is discharged from the
equipment; the concentrate, meanwhile, is
discharged from the bottom. So here we have an
evaporator in which the substance to be treated
flows in the form of a descending film along the
internal walls of a tube bundle heated by steam.
The main advantage of these evaporators is that
they provide a thinner and faster film and
therefore the heat transmission coefficient is
quite high.

Figure 2. Descending film evaporator

2 EXAMPLE OF AN
EVAPORATOR STATION IN
THE SUGAR INDUSTRY

Figure 3 shows a schematic of a typical
installation in a sugar factory.

The schematic shows an evaporator station
comprised of five effects and eight evaporators.
Each of the first five effects is composed of two
Robert type evaporators with parallel flow,
while the last two effects are comprised of two
descending film evaporators. Furthermore, they
also include the associated preheaters of the
sweetened juice feed, expansion balls and a
barometric condenser.

Jugo filtrado

Jarabe a tachas

Tachas

Figure 3. Schematic of an evaporator station

3 MATHEMATICAL MODEL OF
AN EVAPORATOR

The model has been built on the basis of
different hypotheses:

1. In the evaporator we have a perfect
mix.

2. In the concentration chamber the
steam and the solution to be
concentrated are in equilibrium.

3. There is no accumulation of
condensate in the calandria or the
shell.

4. There is no accumulation of heat
on the walls or in the tubes.

5. No consideration is given to the
non-condensates from the steam
condensation chamber.

6. No consideration is given to the
heat losses to the exterior.

The following abbreviations are used:

W mass flow of steam
Wc mass flow of condensate
F mass flow of feed
C mass flow of concentrated juice (syrup)
E vapour from the juice concentration

chamber
c fraction of solids in the juice

m juice mass in the evaporator
mv steam mass in the evaporator
T temperature
P pressure
h juice enthalpy
H steam enthalpy
hc condensate enthalpy
q flow of heat transferred from the

condensation chamber to the juice
concentration chamber

t time

They are all expressed in SI units.

In the case of an evaporator in the sugar
industry which is used to concentrate a
sweetened juice, the total mass, sucrose mass
and energy balances in the juice concentration
chamber are as follows:

CEF
dt
dm

−−= [1]

cCcF

The juice enthalpy, h, is a function of the sugar
content, c, and of the temperature, T.

h=1500 + (4122-1512c)T + (0.55+3.75c)T2 [4]

We can therefore write the follow expression:

and, working with these equations, they can be
expressed more appropriately as:

In this way, the mass, Brix and temperature of
the juice are state variables of the model. The
steam enthalpy is a known function of the
pressure and the temperature:

H = 2.5 106 + 1813 Tsat + 0.471 Tsat2 -  0.011
Tsat3 + 2090 (T - Tsat)      [7]

where Tsat = 2147/(10.76- log(P)) –273.2

and the pressure, p, is fixed by the ratio of
equilibrium:

T= Tsat + a1+ a2 Tsat + a3Tsat2 [8]

where:

a1=-0.2 + exp(-1.5254+2.2962c+2.163c2) [9]

a2=0.01(-0.15exp(-3.202+6.6743c-1.161c2))[10]

a3 = 1e-4 exp(-1.4278 – 2.4382c  + 6.047c2) [11]

So now, to obtain the evolution of m, c and T in
time, we only need additional equations for the
heat and mass flows. Since q depends on the
transmission of heat from the condensation
chamber, the mass and energy balances of this
chamber are expressed as follows:

c
v WW

dt
dm

−= [12]

qhWHW
dt

Hmd
cc

v −⋅−⋅=
)(

[13]

The constants of time associated with these
balances are very low in relation with those of
the concentration chamber. It was therefore
possible to assume a steady state in the
condensation chamber and express the
following:

W·(H-hc) = q [14]

Here, H is a function of P1 and T1, and

hc(T) = 1500 + 1122T + 0.55 T2 [15]

Therefore, beginning with the initial values of
the state variable and for certain external
conditions, all the flows can be determined and,
therefore, by integrating the differential
equation we obtain new values for T, c and m.

In the event that we have multiple effects, the
flow of steam from each evaporator is
calculated taking into account that it is equal to
the flow of steam which condenses in the next
evaporator.

4 EVAPORATION LIBRARY

Based on the models presented, a series of
components and ports has been developed
which forms the evaporation library. The
fundamental components are the evaporators
and their chambers, as well as the preheaters,
balls, etc. In the same way, juice and steam
ports are identified. In all these a series of
functions is used with which different physical
properties are calculated, whether they belong
to the steam or to the solution to be
concentrated.

With all the ports created, the system
automatically introduces all the related
equations with the way in which the
connections are made. This means, basically,
equality of pressure and addition of the mass
flows (taking into account their inlet or outlet
direction) in all the connection points.

PORT vapor

       EQUAL REAL P RANGE 0,Inf       "Presión(bar)"
                     REAL T RANGE -273,Inf "Temperatura (ºC)"
       SUM     REAL W RANGE 0,Inf      "Flujo másico de

vapor (kg/s)"
       SUM     REAL f_energ "Flujo de energía (KW)"
CONTINUOUS

T=temperatura_sat(P)
f_energ=W*entalp_vap(P)

END PORT

PORT jugo
        SUM   REAL W RANGE 0,Inf  "Flujo másico (kg/s)"
                    REAL c  RANGE 0,1     "Fracción de sólidos

(%1 en peso)"
                    REAL T  RANGE -273,Inf "Temperatura (ºC)"
        SUM   REAL  f_energ "Flujo de entalpía (KW)"
                    REAL  F "Flujo volumétrico (m3/s)"

CONTINUOUS
f_energ=W*h_jugo(T,c)

W=F*den_jugo(T,c)
END PORT

Among the constants defined we have the purity
of the solution to be concentrated, because this
value remains constant throughout the complete
evaporation stage.

CONST REAL pureza=0.93

With respect to the components, the following
have been developed:

• Condensation chamber
• Concentration chamber (Robert)
• Concentration chamber

(descending film)
• Evaporator
• Flow mixer
• Flow disperser
• Heat exchangers
• Barometric condenser
• Proportional controller
• PI controller

The following studies the most significant in
more detail.

4.1 CONDENSATION CHAMBER
COMPONENT

This component is common to all types of
evaporators. It basically consists of a chamber
into which steam flows, is condensed and
causes a latent heat transfer, and from which the
same flow, now condensed, is discharged.

This component therefore contains the
equations that refer to the balance of mass and
energy of the chamber. These equations are
represented in steady state for the reasons
indicated in the development of the
mathematical model.

COMPONENT camara_cond

PORTS

IN vapor v_in
OUT agua f_out
OUT termico q_out

CONTINUOUS
v_in.W=f_out.W
v_in.f_energ-f_out.f_energ=q_out.q
f_out.P=v_in.P
q_out.T=f_out.T

END COMPONENT

4.2 CONCENTRATION CHAMBER
COMPONENT

The concentration chamber of an evaporator is
modelled according to the type of model in
question, because each evaporator has different
characteristics which make them particularly
advantageous in determined applications. In
particular, the available surface for heat
transmission varies from some types to others.

The Robert evaporator is a sugar industry
evaporator par excellence, although as we
mentioned earlier, it is being substituted by the
descending film evaporators.

The equations used in this component have
already been described and, as in the previous
case, they basically correspond to the mass and
energy balances and equilibrium equations
related with that chamber.

COMPONENT camara_conc

PORTS
IN termico q_in
IN jugo f_in
OUT jugo f_out
OUT vapor v_out

DATA
REAL Nt=5000 "número de tubos"
REAL di=0.03 "diámetro interior de

los tubos (m)"

DECLS

REAL c -- fracción de sólidos en el
jugo (%Brix/100)

REAL m -- masa de jugo (kg)
REAL T -- temperatura (ºC)
REAL Pjugo --Presión jugo (bar)
REAL dhc      -- derivada de entalpía

respecto a concentración
REAL dhT    -- derivada de la entalpía

respecto a la temperatura

REAL volumen
REAL nivel
REAL Sup   -- Superficie disponible

para la circulación de jugo

REAL SP_L -- consigna de nivel

CONTINUOUS

dhc=(-1522*T+3.75*T**2)/1000

dhT=(4122-1522*c+1.1*T+7.5*T*c)/1000

m'=f_in.W-f_out.W-v_out.W

c'=(f_in.W*(f_in.c-c)+v_out.W*c)/m

T'=(f_in.W*(h_jugo(f_in.T,f_in.c)-
       h_jugo(T,c))-dhc*(f_in.c-c)\
      -v_out.W*(entalp_vap(v_out.P)-
       h_jugo(T,c)-dhc*c)+q_in.q)/(m*dhT)

v_out.P=presion(T,c)

q_in.T=T

f_out.T=T

f_out.c=c

Pjugo=v_out.P+den_jugo(T,c)*9.8*nivel)/
10**5

volumen=m/den_jugo(T,c)  --volumen de jugo

Sup=PI*di**2*Nt/4

nivel=(volumen-4.5)/Sup

END COMPONENT

4.3 EVAPORATOR

Once the components of the condensation and
concentration chambers have been defined, it is
easy to define an evaporator by composition.

USE MATH

USE PROCESS

COMPONENT un_evapR

PORTS

IN jugo jugo_in
IN vapor vapor_in
OUT vapor vapour_out
OUT agua cond_out
OUT jugo jugo_out

DATA

REAL Ntubos=5000   "Número de tubos"
REAL dint =0.033  "diámetro interior tubos(m)"
REAL esp=0.002   "espesor de los tubos(m)"
REAL Long=3.5    "longitud de los tubos(m)"
REAL Usup =2

DECLS
REAL salto

TOPOLOGY
camara_cond c1
camara_conc c2
Pared p1

CONNECT c1.v_in TO vapor_in
CONNECT c1.f_out TO cond_out
CONNECT c2.f_in TO jugo_in
CONNECT c2.f_out TO jugo_out
CONNECT c2.v_out TO vapour_out
CONNECT c1.q_out TO p1.pared_in
CONNECT p1.pared_out TO c2.q_in

CONTINUOUS

salto = cond_out.T - jugo_out.T

END COMPONENT

4.4 PREHEATER COMPONENT

The evaporator feed is preheated in a row of
heat exchangers in series. These are heat
exchangers that are comprised of a shell and
tubes. Through these tubes flows the solution
which is to be heated to the desired temperature
(the temperature of the concentration chamber
of the first evaporator) and through the chamber
flows the steam (normally small vapour
currents) which condenses.

Once the flows, temperatures, pressures... are
known the flow of transferred heat can be
calculated, because the heat given off by steam
condensation is equal to the heat taken by the
sweetened juice and, in turn, it is the heat which
is transferred through the heat exchanger tube
wall. That is:

lnTSU ∆⋅⋅=∆⋅⋅=⋅= TCmmq pjjv λ [16]

where mv, is the mass flow of steam that
condenses, λ the latent condensation heat, mj,
the mass flow of sweetened juice, Cpj, the
specific heat of the juice, ∆T the difference in
temperature between juice output and input,
U the overall heat transmission coefficient,
S the heat exchanger surface and ∆Tln, the
difference in temperature (logarithmic average
between the extremes of the evaporator
temperature profile).

As the flow of transferred heat and the
temperature difference are known, assuming an
overall heat transmission coefficient from the
previous equation, the value of the surface S can
be obtained.

Some standard heat exchanger tube dimensions
are selected, ie, inside diameter, thickness and
length. With these values we can calculate the
surface of each tube and, therefore, the number
of tubes.

LdS otubo ⋅⋅= π [17]

tubo
tubos S

S
N = [18]

Once all the heat exchanger dimensions have
been calculated, the assumed overall heat
transmission coefficient is verified (equation
19). To do this, we have to calculate the
individual heat transmission coefficients on the
tube side, hi, and on the shell side, ho, with the
use of empirical equations.

where Kw is the thermal conductivity of the tube
material hod  and hid the coefficients that take
into account the effect of dirt on the
transmission of heat for the shell and tube sides,
respectively.

Individual heat transmission coefficient on the
outside of the tubes

Complete condensation takes place in the shell.
The overall heat transmission coefficient in the
case of condensation on the outside of
horizontal tubes is calculated for one tube only
and it is corrected with a factor related to the

number of tubes in a horizontal row, expressed
as follows:

where:

KL, thermal conductivity of the liquid, W/m2°C
ρL, density of the liquid, kg/m3

ρV, density of the stem, kg/m3

µL, viscosity of the condensate, kg/m s
ΓV, flow of condensate per perimeter unit of
tube, that is, Wc/ (Nt L)
Nr, average number of tubes per vertical row

Individual heat transmission coefficient on the
inside of the tubes

Sweetened juice flows on the inside of the
tubes, where it is heated by means of
convection. In this case, the heat transmission
coefficient is calculated with equation 21.

33.08.0 PrRe023.0 LL
L

di
k

hi ⋅⋅





⋅=      [21]

where:
kL , thermal conductivity of the liquid, W/m2°C
di, inside diameter of the tube, m
Re, Reynolds number, adimensional=(d u ρ / µ)
Pr, Prandtl number, adimensional=(Cp µ / kL)

Based on the foregoing, we can model the
preheater as a component comprised of another
three components:

1. the shell where steam condensation takes
place with the consequent emission of heat,

2. the tube walls through which heat is
transmitted, and

3. the inside of the tubes through which the
solution to be heated flows.

A steady state has been considered for the
preheaters, in order to reduce the number of
equations and simplify the calculations. In
addition, a constant thermal gradient has been
assumed on the tube side (equivalent to the
minimum thermal gradient), which gives us the
most unfavourable situation.

USE MATH

USE PROCESS

PORT termico_camb
EQUAL REAL Tent
EQUAL REAL Tsal
SUM REAL q

END PORT

COMPONENT camara_liq

PORTS
IN jugo f_in
IN termico_camb q_in
OUT jugo f_out

CONTINUOUS
f_in.W=f_out.W
q_in.Tent=f_in.T
q_in.Tsal=f_out.T
q_in.q=f_out.W*4.21*(f_out.T-f_in.T)

END COMPONENT

COMPONENT camara_conden

PORTS
IN vapor v_in
OUT agua f_out
OUT termico_camb q_out

CONTINUOUS
v_in.W=f_out.W
v_in.f_energ-f_out.f_energ=q_out.q

f_out.P=v_in.P
q_out.Tent=q_out.Tsal
q_out.Tent=f_out.T

END COMPONENT

COMPONENT pared_camb
PORTS

IN     termico_camb pared_in
OUT  termico_camb pared_out

CONTINUOUS
pared_in.q = pared_out.q

END COMPONENT

COMPONENT precalentador
PORTS

IN jugo jugo_in
OUT jugo jugo_out
IN vapor v_in
OUT agua cond_out

DATA
         REAL temp=8 "salto térmico del jugo en

el precalentador(ºC)"

TOPOLOGY
     camara_liq cjugo
     camara_conden            cvapor
     pared_camb pared

     CONNECT jugo_in TO cjugo.f_in
     CONNECT jugo_out TO cjugo.f_out
     CONNECT v_in TO cvapor.v_in
     CONNECT cond_out TO cvapor.f_out
     CONNECT cvapor.q_out TO pared.pared_in
     CONNECT pared.pared_out TO jugo.q_in

CONTINUOUS
cjugo.f_in.c=cjugo.f_out.c
cjugo.f_out.T=cjugo.f_in.T +temp

END COMPONENT

4.5 BAROMETRIC CONDENSER

The steam from the last effect of an evaporate
is usually extracted by means of a condenser.
The most commonly used type is the
countercurrent barometric condenser. This is
direct contact equipment in which steam is
condensed as it ascends through a fine mist of
cooling water. These condensers are not
expensive and prove to be economical as far as
water consumption is concerned.

The barometric condenser is so called because
of the way the condensate is removed. It
descends through a vertical pipe, known as a
barometric head, which is long enough so that
the pressure at the bottom is slightly higher than
atmospheric pressure. The condensate is vented
to a tank which is open to the atmosphere. The
condenser must therefore be installed high
enough so that the water can be discharged by
gravity from the condenser vacuum.

The ratio of water consumption versus the flow
of condensed steam can be determined by
means of the following equation:

12

2

hh
hH

vapordeFlujo
aguadeFlujo v

−
−

=                   [22]

where Hv is the enthalpy of the steam, and h1
and h2 is the enthalpy of the water that enters
and leaves the condenser.

USE PROCESS

COMPONENT cond_bar

PORTS

IN vapor v_in

IN agua f_in

OUT agua f_out

DATA

  REAL Cp=4.184 "calor específico (KJ/kg

ºC)"

DECLS

 REAL sp_P

CONTINUOUS

 f_out.W=f_in.W+v_in.W

f_out.P=v_in.P

f_in.W=v_in.W*(entalp_vap(v_in.P)-

h_liq(f_out.T))/(h_liq(f_out.T)-Cp*f_in.T)

END COMPONENT

5. EXAMPLE OF APPLICATION

As an example the process shown in Figure 3
has been simulated, joining the library
components necessary and adding the essential
control elements; specifically, juice level
regulators in each evaporator and syrup flow
regulators at the outlet. Different experiments
were subsequently carried out, of which some of
the results are given below:

A. Live steam pressure rise

The live steam pressure has risen from 3 to
3.3 bar (Figure 4) approximately at the instant
of time 2500 and, as a consequence, there has
been a rise in the pressure (Figure 5) and
temperature of the concentration chamber
(Figure 6). From these figures we can see the
temporary evolution of these magnitudes in the
different evaporators, according to the legends
in the figures. Likewise, we can see an increase
in live steam flow (Figure 7), in this case in the
first two evaporators because they are in
parallel, and an increase in the feed juice
(Figure 8). We can also see a decrease in the
flow of syrup extracted (Figure 9) and therefore
there are no significant changes in the syrup
concentration (Figure 10).

Figure 4. Rise in live steam pressure

Figure 5.  Vapour pressure

Figure 6. Concentration chamber temperature

Figure 7. Live steam flow

Figure 8. Feed juice flow

Figure 9. Extracted syrup flow

Figure 10. Juice concentration

B. Rise in flow of extracted syrup

Exactly as though we had only one evaporator,
as the discharge flow of final syrup increases
(Figure 11) the less amount of time the juice
remains in the evaporator and, therefore, the
syrup concentrations at the outlet decrease
(Figures 12 and 13). Because of this, to
maintain the level of juice in the tubes the feed
flow must be increased (Figure 14). With the
addition of more juice, the concentration
continues to decrease and it is therefore
necessary to increase the flow of steam
(Figure 15).

Figure 11. Flow of syrup at the outlet of the
evaporators

Figure 12. Concentration of syrup at the outlet
of the first and second effects

Figure 13. Concentration of syrup at the outlet
of the remaining evaporators

Figure 14. Flow of feed juice

Figure 15. Flow of live steam

6 CONCLUSIONS

Throughout this work, one of the main
advantages of EcosimPro has been put to good
use: the modularity. This means that once we
have the dynamic model of an evaporator and
its correct operation has been checked, it forms
a component which can be connected to other
components, whether they are the same or not.
In this way we can “build” any evaporator
station, no matter how many evaporators it has
or their arrangement (in series, in parallel, etc),
using a minimum number of components.

The above results show great similarity with the
real behaviour of evaporator stations. This
means we have an evaporation library in
EcosimPro which can be used as a reliable tool
for the simulation of different operating
conditions.

